

## ПРАКТИКУМ 5.2.3. ВСТРОЕННЫЕ ФУНКЦИИ ТАБЛИЧНОГО ПРОЦЕССОРА

<b>ПРАКТИКУМ 5.2.3. ВСТРОЕННЫЕ ФУНКЦИИ ТАБЛИЧНОГО ПРОЦЕССОРА EXCEL.....</b>	<b>1</b>
<b>ОБЩИЕ СВЕДЕНИЯ О ФУНКЦИЯХ .....</b>	<b>1</b>
<i>Синтаксис встроенной функции .....</i>	<i>1</i>
<i>Вставка встроенной функции .....</i>	<i>2</i>
<i>Правила построения формул с помощью Мастера функций.....</i>	<i>4</i>
<b>КАТЕГОРИИ ФУНКЦИЙ .....</b>	<b>4</b>
<i>Математические функции.....</i>	<i>5</i>
<i>Функции категории Дата и время .....</i>	<i>7</i>
<i>Функции категории Ссылки и массивы .....</i>	<i>7</i>
<b>СТАТИСТИЧЕСКИЕ ФУНКЦИИ.....</b>	<b>8</b>
<i>Общая характеристика статистических функций .....</i>	<i>8</i>
<i>Статистические и математические функции с условиями.....</i>	<i>10</i>
<b>ЛОГИЧЕСКИЕ ФУНКЦИИ.....</b>	<b>11</b>
<i>Простые логические выражения.....</i>	<i>11</i>
<i>Функция ЕСЛИ ().....</i>	<i>11</i>
<i>Составные логические выражения.....</i>	<i>14</i>
<i>Использование вложенных функций ЕСЛИ() .....</i>	<i>15</i>
<b>ЭЛЕМЕНТЫ ИНТЕРФЕЙСА EXCEL 2007 ДЛЯ РАБОТЫ С ФУНКЦИЯМИ.....</b>	<b>18</b>
<b>ПРАКТИЧЕСКИЕ ЗАДАНИЯ.....</b>	<b>19</b>
<i>Задание 1. Вкладка Формулы и справочная система по функциям.....</i>	<i>19</i>
<i>Задание 2. Штраф за отходы .....</i>	<i>20</i>
<i>Задание 3. Фитнес-центр .....</i>	<i>21</i>
<i>Задание 4. Сравнение вкладов .....</i>	<i>23</i>

### **Общие сведения о функциях**

#### **Синтаксис встроенной функции**

Важным инструментом обработки данных в табличных процессорах являются встроенные функции.

Функции выполняют вычисления по запрограммированным алгоритмам. Исходными данными для вычислений являются заданные различным способом величины, называемые аргументами. Функции позволяют выполнять как простые, так и сложные вычисления.

Каждая встроенная функция имеет имя и список аргументов, которые записываются в скобках и разделяются знаком «точка с запятой».

ИМЯ\_ФУНКЦИИ (список аргументов)

Например, функция для возведения числа в степень имеет следующий вид:

СТЕПЕНЬ (<число>; <степень>),

где СТЕПЕНЬ – это имя функции, а <число> и <степень> – это список аргументов, означающий что возводить и в какую степень.

Аргументами функций могут быть данные разного типа: константы, арифметические и логические выражения, выражения, содержащие ссылки на ячейки или диапазоны ячеек. Диапазоны ячеек могут быть заданы именами. Выражения могут содержать обращения к другим функциям.

Для примера, приведенного выше, вызываемая в Excel функция может иметь следующий вид:

СТЕПЕНЬ (3; 2) – аргументы заданы константами;

СТЕПЕНЬ (A1; B1) - аргументы заданы относительными ссылками на ячейки таблицы A1 и B1;

СТЕПЕНЬ (C3; \$A\$1) - аргументы заданы относительной ссылкой на ячейку C3 таблицы и абсолютной ссылкой - на ячейку A1;

СТЕПЕНЬ (СУММ(C3:C10); \$A\$1) – первый аргумент задан вложенной функцией суммирования чисел в диапазоне C3:C10 таблицы (фактически это выражение, содержащее функцию) и абсолютной ссылкой на ячейку A1.

## Вставка встроенной функции

Вставить встроенную функцию можно вызвать одним из следующих способов:

- выполнить команду **Формулы / Библиотека функций / Вставить функцию**  ;
- нажать кнопку **Вставка функции** , расположенную слева от строки формул;
- с помощью комбинации клавиш **Shift + F3**;
- выбрать последнюю команду **Другие функции** в любой категории функций в группе **Библиотека функций**;
- выбрать строку **Другие функции** в раскрывающемся поле **Имя** слева от строки ввода формул (рис. 1)






Рис. 1. Список функций в поле Имя

Команда запускает **Мастер функций**, работа которого состоит из двух шагов.



**Рис. 2. Выбор категории и функции**

На первом шаге необходимо в появившемся диалоговом окне выбрать нужную категорию из списка категорий (рис. 2) и по категории найти функцию. Если вы не знаете ни категории, ни названия функции, то необходимо в поле Поиск (рис. 3) функции ввести краткое описание ваших действий или найти подходящую функцию в полном алфавитном перечне.



**Рис. 3. Поиск функции**

После выбора функции под списком функций можно увидеть формат функции, краткое описание ее действий и гиперссылку на справку по этой функции.

На втором шаге мастера в диалоговом окне Аргументы функции необходимо ввести все требуемые аргументы. Каждый аргумент вводится в специально

предназначенное для него поле. Поля для ввода аргументов могут быть обязательными для заполнения и необязательными. Первые выделяются полужирным шрифтом. При вводе аргументов рядом с полем ввода появляется их значение и во второй части окна - значение всей формулы (рис. 4).




Рис. 4. Окно Аргументы функций

После завершения второго шага Мастера функций в ячейке с функцией появится ее значение.

### Правила построения формул с помощью Мастера функций

1. Состав аргументов функций, порядок их следования и типы значений фиксированы и не подлежат изменению.
2. Аргументы вводятся в специальных полях ввода.
3. Для вызова вложенной функции используется список функций в поле **Имя**.
4. Для ввода имени блока ячеек используется команда **Формулы / Определенные имена / Использовать в формуле**.
5. Для построения ссылки надо сначала щелкнуть в поле ввода , а выделить ячейку или диапазон.
6. Чтобы изменить тип адреса ячейки, надо щелкнуть внутри ссылки и нажать клавишу F4 один или несколько раз.

### Категории функций

Все многообразие функций делятся на категории по назначению:

- математические;
- статистические;
- логические;
- текстовые;
- дата и время;
- ссылки и массивы;

- финансовые и др.

Категория **Полный алфавитный перечень** содержит функции всех категорий в алфавитном порядке: сначала с англоязычными названиями от А до Z, затем с русскоязычными – от А до Я.

Очень удобна категория **10 недавно использовавшихся**.

Рассмотрим подробнее содержимое некоторых категорий.

## Математические функции

Список часто используемых математических функций приведен в таблице (таблица 1).

**Таблица 1. Некоторые функции категории Математические**

Функция	Описание функции	Примеры записи
СУММ(число1;число2;...)	Суммирует значения аргументов. Нечисловые значения игнорируются.	СУММ (A1:A5) СУММ (Прибыль) СУММ (A1;B2;C4) СУММ(A1;Прибыль;15)
КОРЕНЬ(число)	Вычисляет квадратный корень числа.	КОРЕНЬ(5) КОРЕНЬ (A1) КОРЕНЬ(B1*B1-4*A1*C1)
ОКРУГЛ(число; число разрядов)	Округляет число до указанного числа разрядов	ОКРУГЛ(3,1415; 2) ОКРУГЛ(A1; B1)
СТЕПЕНЬ (число; степень)	Возводит число в указанную степень	СТЕПЕНЬ(5; 3) СТЕПЕНЬ(A1; B1)
ЦЕЛОЕ(число)	Округляет число до ближайшего меньшего целого	ЦЕЛОЕ(5,123) ЦЕЛОЕ(A1)
ABS(число)	Вычисляет модуль числа	ABS(-5) ABS(A1)
ОСТАТ(число; делитель)	Вычисляет остаток от деления нацело	ОСТАТ(17;4) ОСТАТ((B4-C4)*D4;A1)
SIN(угол)	Вычисляет синус угла, заданного в радианах	SIN(1,5) SIN(A1) SIN(ПИ()/4)

Тип результата зависит от типа аргументов и от особенностей работы самой функции. Так результатом вычисления функции ЦЕЛОЕ() будет целое число, а результат работы функции ABS() зависит от типа аргумента.

На рис. 5 приведены примеры использования некоторых математических функций.

	A	B	C	D	E
1					
2	<b>Отображение формул с математическими функциями</b>				
3		ЧИСЛО	КОРЕНЬ (число)	ОКРУГЛ (число; число разрядов)	СТЕПЕНЬ (число; степень)
4		0,25	=КОРЕНЬ(B4)	=ОКРУГЛ(C4;1)	=СТЕПЕНЬ(B4;1)
5		3,1415	=КОРЕНЬ(B5)	=ОКРУГЛ(C5;2)	=СТЕПЕНЬ(B5;2)
6		2	=КОРЕНЬ(B6)	=ОКРУГЛ(C6;3)	=СТЕПЕНЬ(B6;3)

	A	B	C	D	E
1					
2	<b>Результаты вычислений по формулам</b>				
3		ЧИСЛО	КОРЕНЬ (число)	ОКРУГЛ (число; число разрядов)	СТЕПЕНЬ (число; степень)
4		0,25	0,5000	0,5	0,25
5		3,1415	1,7724	1,77	9,87
6		2	1,4142	1,414	8,00

**Рис. 5. Примеры использования математических функций**

Математическая функция СЛЧИС() не имеет аргументов и возвращает равномерно распределенное случайное число на отрезке [0; 1) При новом обращении к функции и при нажатии клавиши F9 генерируется новое число. Пример работы функции СЛЧИС() приведен на рис. 6.

	A	B	C	D	E	F	G
1				Нижняя граница диапазона	5	Верхняя граница диапазона	15
2	Действительные числа в интервале от 0 до 1	Действительные числа в интервале от 0 до 5	Целые числа в диапазоне от 10 до 15	Целые числа в диапазоне от E1 до G1			
3	0,66	1,18	13	15			
4	0,10	1,88	10	8			
5	0,90	0,23	13	12			
6	0,29	1,69	10	15			
7	0,02	2,07	15	12			
8	0,72	4,41	15	11			
9	0,58	2,78	15	9			
10	0,28	0,34	14	8			
11							

**Рис. 6. Примеры использования функции СЛЧИС()**

Для получения действительных чисел в интервале от 0 до 1 (столбец A) используется формула =СЛЧИС().

Для получения действительных чисел в интервале от 0 до 5 (столбец B) используется формула =5\*СЛЧИС().

Для получения целых чисел в диапазоне от 10 до 15 (столбец C) используется формула =ЦЕЛОЕ(СЛЧИС()\*6)+10.

Для получения целых чисел в произвольном диапазоне (столбец D), границы которого находятся в ячейках E1 и G1, используется формула  

$$= \text{ЦЕЛОЕ}(\text{СЛЧИС()} * (\$G\$1 - \$E\$1 + 1)) + \$E\$1.$$

В данном примере числа будут генерироваться в диапазоне от 5 до 15.

## Функции категории Дата и время

Функции этой категории работают со специфическими числовыми масштабами даты и времени (таблица 2).

**Таблица 2. Некоторые функции категории Дата и время**

Функция	Описание функции	Примеры записи
СЕГОДНЯ ()	Возвращает текущую дату в формате даты. Если до ввода этой функции для ячейки был задан формат Общий, результат будет отформатирован как дата.	СЕГОДНЯ ()
ГОД (дата)	Возвращает год, соответствующий заданной в качестве аргумента дате. Год определяется как целое число в диапазоне от 1900 до 9999.	ГОД (15.03.2009) ГОД (A3)
МЕСЯЦ (дата)	Возвращает месяц для даты, заданной в качестве аргумента числовом формате. Месяц возвращается как целое число в диапазоне от 1 (январь) до 12 (декабрь).	МЕСЯЦ (15.03.2009) МЕСЯЦ (C6)
ДАТА (год; месяц; день)	Возвращает число, представляющее определенную дату. Если до ввода этой функции формат ячейки был задан как Общий, результат будет отформатирован как дата.	ДАТА (2004;08;25) ДАТА (A4;B4;C4)
ДЕНЬ(дата)	Возвращает число месяца от 1 до 31	ДЕНЬ(27.08.2005) ДЕНЬ(C6)
КОНМЕСЯЦА (дата; число)	Возвращает дату последнего дня месяца, отстоящего на указанное количество месяцев от даты, указанной в аргументе «нач_дата».	КОНМЕСЯЦА(A3;-1) КОНМЕСЯЦА (СЕГОДНЯ(); 0)

## Функции категории Ссылки и массивы

Функции этой категории могут быть полезны для определения местоположения данных для расчетов в таблице (таблица 3).

**Таблица 3. Некоторые функции категории Ссылки и массивы**

Функция	Описание функции	Примеры записи
ВЫБОР (номер_индекса; значение1; значение2;...)	Возвращает значение из списка аргументов-значений по заданному номеру индекса. Функция ВЫБОР позволяет выбрать одно значение из списка, в котором может быть до 254 значений.	ВЫБОР (3;A1;A2;A3)
ПОИСКПОЗ (искмое_значение; просматриваемый_массив; тип_сопоставления)	Возвращает относительное положение элемента массива (позицию элемента в диапазоне), который соответствует указанному значению в указанном порядке.	ПОИСКПОЗ (50;A1:A7;0)
ГПР (искмое_значение; таблица; номер_строки; интервальный_просмотр)	Функция горизонтального просмотра выполняет поиск значения в верхней строке таблицы или массива значений и возвращает значение того же столбца в заданной строке таблицы или массива.	ГПР(7;A1:B12; 2;ЛОЖЬ)

## Статистические функции

### Общая характеристика статистических функций

Статистические функции занимают особое место при решении экономических задач и имеют столь широкое применение, что требуют более подробного рассмотрения.

К категории Статистические (рис. 7) относятся следующие часто используемые функции, представленные ниже (таблица 4).

Таблица 4. Некоторые функции категории Статистические

Функция	Описание функции	Примеры записи
МАКС(число1; число2;...)	Находит максимальное значение из списка аргументов. Нечисловые значения игнорируются	МАКС(A1:A5) МАКС(Прибыль) МАКС(A1;B2;C\$3)
МИН(число1; число2;...)	Находит минимальное значение из списка аргументов. Нечисловые значения игнорируются	МИН(A1:A5) МИН(A1;B2;C\$3) МИН(A1; Прибыль;15)
СРЗНАЧ(число1;число2;...)	Находит среднее значение из списка аргументов. Нечисловые значения игнорируются	СРЗНАЧ (A1:A5) СРЗНАЧ (Прибыль) СРЗНАЧ (A1;B2;C\$3)
СЧЕТ(значение1; значение2;...)	Подсчитывает количество чисел в списке аргументов. Нечисловые значения игнорируются	СЧЕТ (A1:A5) СЧЕТ (Прибыль) СЧЕТ (A1;B1;C1;4) СЧЕТ (“да”;B1;”нет”;4)
СЧЕТЗ(значение1; значение2;...)	Подсчитывает количество значений в списке аргументов независимо от их типа.	СЧЕТЗ (A1:A5) СЧЕТЗ (Прибыль) СЧЕТЗ (A1;B1;C1;12)

Вид функции с перечнем аргументов дается в интерпретации Excel 2007.




Рис. 7. Статистические функции

Аргументами представленных функций могут быть числа, ссылки на ячейки, имена диапазонов и диапазоны. Если аргументы содержат нечисловую информацию, то они при вычислениях заменяются на 0. Результатом работы функций являются числа.

На рис. 8 приведен пример использования функции МАКС () для определения максимальной прибыли. Аргументом здесь является диапазон значений прибыли.


D3		fx =МАКС(C3:C6)		
	A	B	C	D
1				
2		Квартал	Прибыль	Максимальная прибыль
3		1	3900	4350
4		2	4350	
5		3	2390	
6		4	3100	

Рис. 8. Функция МАКС ()

E3		fx =СЧЁТ(D3:D6)			
	A	B	C	D	E
1					
2			Квартал	Прибыль	Число кварталов с прибылью
3			1	4500	2
4			2	нет прибыли	
5			3	2300	
6			4	нет прибыли	
7					

Рис. 9. Функция СЧЕТ()

Функция СЧЕТ () определяет число ячеек с числами, а функция СЧЕТЗ () - число непустых ячеек. На рис. 9 приведен пример использования функции СЧЕТ () для определения числа кварталов, за которые была получена прибыль. Аргументом функции является диапазон значений, содержащий числовую и текстовую информацию. Результатом применения к этому диапазону функции СЧЕТ () будет число 2, а функции СЧЕТЗ () – число 4.



Строка состояний

**Рис. 10. Отображение статистики в строке состояний**

Чтобы узнать значение наиболее используемых статистических функций СУММ(), МИН(), МАКС(), СРЗНАЧ() и некоторых других достаточно выделить диапазон ячеек и, щелкнув правой кнопкой мыши на строке состояний, в контекстном меню выбрать нужные функции (рис. 10).

Все значения выбранных функций будут отображаться как в самом меню, так и в строке состояний. Такой же набор общеупотребительных статистических функций можно открыть, если выполнить команду **Формулы / Библиотека функций / Автосумма**. В этом случае результат вычисления значения функции появится в ячейке под диапазоном с выделенными ячейками.

### Статистические и математические функции с условиями

Особое место среди статистических и математических функций занимают функции, анализирующие условия при осуществлении расчетов (таблица 5).

**Таблица 5. Функции, осуществляющие счет**

Функция	Описание функции	Примеры записи
СРЗНАЧЕСЛИ (диапазон; условие; диапазон_усреднения)	Возвращает среднее значение всех ячеек в диапазоне, которые соответствуют данному условию. Пустые ячейки игнорируются. Если 3-его аргумента нет - диапазон усреднения равен диапазону, указанному 1-ым аргументом.	СРЗНАЧЕСЛИ (B1:B12;">0") СРЗНАЧЕСЛИ(ПОЛ;"ж";ОКЛАД)
СЧЕТЕСЛИ (Диапазон; критерий)	Подсчитывает количество непустых ячеек в Диапазоне, удовлетворяющих критерию.	СЧЕТЕСЛИ (C1:C5;">0") СЧЕТЕСЛИ (Рост;"<170")
СУММЕСЛИ (диапазон; условия; диапазон_суммирования)	Суммирует ячейки заданного диапазона, соответствующие условиям.	СУММЕСЛИ (Числа;">0") СУММЕСЛИ (Должность;"*адм*";Оклад)

Функция	Описание функции	Примеры записи
	Пустые ячейки и ячейки, содержащие текстовые значения, не учитываются. 3-ий аргумент может отсутствовать.	

## Логические функции

Категория Логические функции содержит функции, позволяющие придавать экономическим расчетам более интеллектуальный характер.

К категории Логические относятся следующие функции ЕСЛИ (), И (), ИЛИ (), НЕ(), которые в качестве аргументов имеют логическое выражение, и две логические константы ИСТИНА () и ЛОЖЬ ().

## Простые логические выражения

*Логическое выражение* – это любое высказывание, рассматриваемое с точки зрения его истинности. Логическое выражение может принимать два логических значения ИСТИНА или ЛОЖЬ.

В среде Excel можно использовать простые и составные логические выражения.

Простое логическое выражение содержит только одну операцию отношения:

> (больше);

< (меньше);

= (равно);

<> (не равно);

>= (больше или равно);

<= (меньше или равно).

Примеры простых логических выражений с операндами разного типа представлены в таблице (таблица 6).

**Таблица 6. Примеры логических выражений**

Тип используемых данных	Пример выражения
Числовой тип	A1>0
	B2-A2<=C\$2
	A1>CPЗНАЧ(A1:A6)
Логический тип	A1=ИСТИНА
	B1<>ЛОЖЬ
	C1 (ячейка C1 содержит данные логического типа)
Текстовый тип	A1="да"
	Пол <> "м"
	Оценка="Отлично"

Выражение A1="да" будет иметь значение ИСТИНА только, если в ячейке A1 записано слово «да» с точностью до регистра букв, и значение ЛОЖЬ, если в ячейке A1 находится любое другое текстовое значение.


## Функция ЕСЛИ ()

Функция ЕСЛИ () используется для проверки выполнения некоторого условия и имеет следующий формат:

ЕСЛИ (логическое выражение; формула\_если\_истина; формула\_если\_ложь).  
 Будем также использовать более короткую запись:

ЕСЛИ (условие; формула 1; формула 2).

Значение функции вычисляется по формуле 1, если логическое выражение принимает значение ИСТИНА, в противном случае значение вычисляется по формуле 2. Чтобы удобнее было определять значение логического выражения, его удобнее формулировать в вопросительной форме.



**Рис. 11. Схематическое представление работы функции ЕСЛИ()**

Результат работы функции возвращается в ячейку с формулой.

Схематически работа функции ЕСЛИ () представлена на рис. 11:

На рис. 12 приведен пример использования функции ЕСЛИ () для расчета прибыли в зависимости от дохода и расхода.

Можно составить несколько равнозначных логических выражений для проверки наличия прибыли. Прибыль есть, если Доход больше, чем Расход ( $\text{Доход} > \text{Расход}$ ), или разность между доходом и расходом больше нуля ( $(\text{Доход} - \text{Расход}) > 0$ ). Выберем первое условие. Если условие выполняется, то прибыль равна разности между доходом и расходом (Формула 1), а если не выполняется, то результатом является сообщение «нет прибыли» (Формула 2)

Формула с использованием функции ЕСЛИ() в общем виде будет выглядеть так:

ЕСЛИ(Доход>Расход; Доход - Расход; «нет прибыли»).

Для первой ячейки вычисляемого столбца (D3) формула имеет вид:

ЕСЛИ(B3>C3; B3 - C3; «нет прибыли»).

	A	B	C	D	E	F
1						
2		Доход	Расход	Прибыль		
3		3600	2400	1200		
4		2500	2660	нет прибыли		
5		4500	3100	1400		
6		3600	3600	нет прибыли		
7						

**Рис. 12. Расчет прибыли**

Формула с функцией ЕСЛИ() сначала вводится в ячейку D3, а затем копируется в другие ячейки столбца. Рассмотрим, как работает формула при копировании (таблица 7).

**Таблица 7. Работа формулы при копировании**

Логическое выражение	Значение логического выражения	Формулы столбца D	Результат выполнения функции
B3>C3	ИСТИНА	=ЕСЛИ(B3>C3;B3-C3;"нет прибыли")	1200
B4>C4	ЛОЖЬ	=ЕСЛИ(B4>C4;B4-C4;"нет прибыли")	нет прибыли
B5>C5	ЛОЖЬ	=ЕСЛИ(B5>C5;B5-C5;"нет прибыли")	1400

Видно, что в 3-ей строке электронной таблицы (рис. 12) доход больше расхода, т.е. условие B3>C3 равно ИСТИНА, поэтому функция вычисляет значение, соответствующее Формуле 1 как B3-C3 и возвращает (выводит) это значение в качестве результата своей работы в ячейку с формулой D3.

В 4-ой строке таблицы доход меньше расхода, т.е. условие B4>C4 не выполняется и равно ЛОЖЬ, поэтому функция вычисляет значение, соответствующее Формуле 2, т.е. выводит сообщение «нет прибыли».

Рассмотрим еще одну ситуацию, в которой используется функция ЕСЛИ() с простым условием.

На рис. 13 приведен пример создания списка кварталов с прибылью. Наличие прибыли выражается с помощью текстовой константы «да», отсутствие - «нет». В задаче требуется вывести номер квартала, если прибыль есть и ничего не выводить, если нет прибыли. Здесь тоже только две возможности.

D3		fx =ЕСЛИ(C3="да";B3;" ")			
	A	B	C	D	E
1					
2		Квартал	Прибыль	Кварталы с прибылью	
3		1	да	1	
4		2	нет		
5		3	нет		
6		4	да	4	

**Рис. 13. Использование текстовых констант**

D3		fx =ЕСЛИ(C3="да";B3)			
	A	B	C	D	E
1					
2		Квартал	Прибыль	Кварталы с прибылью	
3		1	да	1	
4		2	нет	ЛОЖЬ	
5		3	нет	ЛОЖЬ	
6		4	да	4	
7					

**Рис. 14. Неполная форма ЕСЛИ()**

Если не вводить значение аргумента, когда логическое выражение ложно (Формула 2), то во всех ячейках, соответствующих этой ситуации появится значение ЛОЖЬ (рис. 14).

Чтобы этого не происходило, рекомендуется в качестве значения аргумента вводить текстовые константы "" (пусто) или " " (пробел). Текстовые константы записываются в кавычках.

Формула с использованием функции ЕСЛИ() в общем виде будет выглядеть так:

ЕСЛИ(Прибыль="да";Номер квартала; " ")

Рассмотрим несколько строк электронной таблицы на рис. 13.

Логическое выражение для 3-ей строки С3="да" ("да"="да") равно ИСТИНА, поэтому функция вычисляет значение, соответствующее Формуле 1 и возвращает (выводит) номер квартала в качестве результата своей работы в ячейку с формулой D3.

Логическое выражение для 4-ой строки С4="да" ("нет"="дам) равно ЛОЖЬ, поэтому функция вычисляет значение, соответствующее Формуле 2, т.е. выводит пробел.

Аналогично происходит и в других ячейках электронной таблицы.

## Составные логические выражения

Составные логические выражения создаются на основе простых с помощью логических функций И(), ИЛИ(), НЕ().

### Функция И()

Формат функции И():

И(логическое выражение1;логическое выражение2;...)

Функцию И() называют логическим умножением.

Функция И() принимает значение ИСТИНА, если все логические выражения-аргументы имеют значение ИСТИНА.

На рис. 15 приведен пример создания списка студентов, успешно сдавших сессию, которая включает три экзамена. В таблице используются следующие обозначения: положительные оценки – 3, 4, 5, неудовлетворительная оценка – 2, неявка на экзамен – 0.

	F3	=ЕСЛИ(И(С3>2;D3>2;E3>2);B3;" ")				
	A	B	C	D	E	F
1						
2		ФИО студента	Информатика	Математика	Философия	Без двоек
3		Малинин И.П.	3	3	2	
4		Каверин А.Б.	4	5	4	Каверин А.Б.
5		Соснин А.А.	5	4	4	Соснин А.А.
6		Барыкин Н.В.	4	4	3	Барыкин Н.В.
7						

Рис. 15. Студенты, сдавшие экзамены

Считается, что студент сдал сессию, если он не имеет двоек и неявок, т.е. и оценка по информатике больше 2, и оценка по математике больше 2, и оценка по философии больше 2. Можно по-другому записать выражение «студент сдал сессию», когда оценка по всем предметам больше 2. Чтобы обеспечить проверку выполнения всех трех условий, необходимо использовать функцию И().

Формула в общем виде будет выглядеть так:

ЕСЛИ(И(Информатика>2; Математика >2; Философия >2); Фамилия; "" ).

Данная формула имеет составное логическое условие. При вводе формулы с составными условиями необходимо пройти несколько этапов.

1. Вызвать функцию ЕСЛИ().

2. Перейти в поле **Лог\_выражение** окна Аргументы функции ЕСЛИ().
3. Вызвать функцию И() - ее диалоговое окно **Аргументы функции**;
4. Последовательно ввести все аргументы функции И().
5. Вернуться в окно **Аргументы функции** для функции ЕСЛИ(), для чего, не нажимая ОК, щелкнуть мышкой в строке формул после последней закрывающейся скобки.
6. Последовательно ввести оставшиеся аргументы функции ЕСЛИ().
7. Нажать ОК.

### Функция ИЛИ()

Формат функции ИЛИ():

ИЛИ(логическое выражение1;логическое выражение2;...)

Функцию ИЛИ() называют логическим сложением.

Функция ИЛИ() принимает значение ИСТИНА, если хотя бы один из ее аргументов равен ИСТИНА. Функция ИЛИ() принимает значение ЛОЖЬ, если все ее аргументы принимают значение ЛОЖЬ.

На рис. 16 приведен пример создания списка студентов, имеющих 5 хотя бы по одному из предметов, т.е. или оценка по информатике равна 5, или оценка по математике равна 5, или оценка по философии равна 5.

Формула будет содержать составное логическое выражение с функцией ИЛИ () и в общем виде будет выглядеть так:

ЕСЛИ(ИЛИ(Информатика=5; Математика=5; Философия=5); Фамилия; « »)

На рис. 16 приведена формула для ячейки F6, в которой она получена путем копирования из ячейки F3.

	A	B	C	D	E	F
1						
2		ФИО студента	Информатика	Математика	Философия	С пятерками
3		Малинин И.П.	3	3	2	
4		Каверин А.Б.	4	3	4	
5		Соснин А.А.	5	5	5	Соснин А.А.
6		Барыкин Н.В.	4	4	5	Барыкин Н.В.
7						

Рис. 16. Функция ИЛИ() в условии функции ЕСЛИ()


### Использование вложенных функций ЕСЛИ()

Ранее рассматривались примеры, в которых путь решения выбирался в зависимости от двух ситуаций: выполняется условие или не выполняется условие. И, если условие выполнялось, то вычисления производились по Формуле 1, а иначе – по Формуле 2. При этом условия могли быть как простыми, так и сложными.

Очень часто приходится решать задачи с множественным выбором, когда путь решения задачи выбирается в зависимости от трех и более ситуаций. При решении таких задач используются вложенные функции ЕСЛИ() – в качестве аргумента (аргументов) одной функции ЕСЛИ() используются другие функции ЕСЛИ() .

Рассмотрим задачу вычисления процента премии в зависимости от стажа работы сотрудника.

Если стаж меньше 10 лет, то сотрудник получает 30% от базовой премии. Если стаж больше либо равен 10 годам, но меньше 25 лет, то сотрудник получает 60% от базовой премии. В остальных случаях сотрудник получает всю базовую премию. Схематично на рис. 17 представлена числовая ось со значениями стажа, которая разбита на три отрезка: до 10 лет, от 10 до 25 лет и свыше 25 лет.



**Рис. 17. Выбор из 3-х ситуаций**


На первый взгляд может показаться, что надо проверить три условия

Стаж < 10 (простое)

$10 \leq \text{Стаж} < 25$  (составное)


Стаж  $\geq 25$  (простое).

Однако, на логической схеме (рис. 18) видно, что на самом деле достаточно проверить выполнение только двух условий. Как видно из схемы, если мы проверили условие стаж < 10, то проверять условие стаж  $\geq 10$  уже не нужно. А если мы проверили условие стаж < 25, то проверять условие стаж  $\geq 25$  тоже не нужно.



**Рис. 18. Блок-схема выбора из 3-х ситуаций**

В общем виде схема проверки двух условий выглядит так (рис. 19).



**Рис. 19. Проверка 2-х условий**

Эта логическая схема записывается в виде формулы со сложными функциями ЕСЛИ ():


ЕСЛИ(Условие 1; Формула 1; ЕСЛИ(Условие 2; Формула 2; Формула 3))

Аналогичным образом можно включать в формулу проверку трех и более условий.

Для вычисления премии в зависимости от стажа формула в общем виде будет выглядеть так:

ЕСЛИ(Стаж < 10; 30%\*БП; ЕСЛИ(СТАЖ < 25; 60%\*БП; БП)),

где БП – базовая премия.



**Рис. 20. Окно Аргументы функции 1-ой функции ЕСЛИ()**

На рис. 21 приведен пример расчета премии для нескольких сотрудников. На листе представлены исходные данные: размер базовой премии (ячейка С4), процент премии в зависимости от стажа (ячейки С6:С8), список сотрудников и их стаж работы. В строке формул Приведена формула для вычисления премии для сотрудника Ананьева. В столбце D приведены результаты вычисления по формуле.

D11	=ЕСЛИ(С11<10;С\$5*С\$4;ЕСЛИ(С11<25;С\$7*С\$4;С\$4))					
	A	B	C	D	E	F
3						
4		Базовый размер премии	10 000р.			
5						
6		Стаж меньше 10 лет	30%			
7		Стаж от 10 до 25 лет	60%			
8		Стаж от 25 лет и выше	100%			
9						
10		ФИО	Стаж работы	Премия		
11		Ананьев И.М.	18	6000		
12		Борисов К.М.	3	3000		
13		Володин А.В.	15	6000		
14		Голубева М.И.	6	3000		
15		Данилов С.Ю.	27	10000		

**Рис. 21. Расчет премии**

При решении задачи с вложенными ЕСЛИ() ввод формулы, представленной на рис. 21 **Error! Reference source not found.**, осуществляется за несколько шагов.

1. Выбрать функцию ЕСЛИ().
2. В окне Аргументы функции первой функции ЕСЛИ() и ввести значения в поля Лог\_выражение и Значение\_если\_истина
3. Щелкнуть в поле Значение\_если\_ложь, выбрать новую функцию ЕСЛИ.
4. В окне Аргументы функции для вложенной функции ЕСЛИ() () ввести значения в поля Лог\_выражение и Значение\_если\_истина, Значение\_если\_ложь новой функции ЕСЛИ().
5. Закончить работу с Мастером, нажав ОК.

Для корректировки значений аргументов любой функции достаточно перейти в строку формул в поле нужной функции и нажать кнопку .

### **Элементы интерфейса Excel 2007 для работы с функциями**

Для работы со встроенными функциями в Excel 2007 используется вкладка Формулы.

На ней расположены 4 группы: Библиотека функций, Определенные имена, Зависимости формул, Вычисление.

Группа Библиотека функций служит для вызова Мастера функций и для вызова различных категорий функций: математических, логических, статистических и пр.

Группа Определенные имена предназначена для работы с именами ячеек и диапазонов ячеек. Использование имен в формулах делает формулы более понятными. К данным, имеющим имена, можно обращаться по этим именам. Использование имени равносильно абсолютной ссылке: при копировании формул имя не изменяется.

Группа Зависимости формул позволяет управлять отображением или скрытием формул с целью обеспечения безопасности или по каким-либо другим причинам. Для проверки формул можно вывести на экран все формулы листа, а

также стрелки, указывающие на влияющие и зависимые ячейки в данной формуле.

Группа **Вычисление** позволяет изменять параметры вычислений и перерасчетов.

## **Практические задания**

### **Задание 1. Вкладка Формулы и справочная система по функциям**

Освоить возможности справочной системы при работе со встроенными функциями.

#### **Технология выполнения задания**

1. Откройте программу MS Excel.
2. Перейдите на вкладку **Формулы** и поочередно ознакомьтесь с содержимым различных категорий в группе **Библиотека функций**.
3. Заполните ячейки A2:J2 числами от -14 до 13, используя автозаполнение: **Главная/Редактирование/Заполнить/Прогрессия**.
4. В ячейке K2 посчитайте сумму значений ячеек A2:J2, используя команду **Формулы / Автосумма**.
5. В ячейке L2 посчитайте среднее значение ячеек A2:J2, выбрав из списка **Автосумма** функцию **СРЗНАЧ()**. Диапазон ячеек A2:K2 замените на A2:J2.
6. Найдите в группе **Библиотека функций** категорию **Математические**.
7. Для подсчета суммы отрицательных чисел в диапазоне A2:J2 в ячейку M2 вставьте функцию **СУММЕСЛИ()**.
8. Введите в поле аргумента **Диапазон** исходный диапазон, выделив мышью ячейки A2:J2.
9. Введите в поле аргумента **Критерий** значение "<0". При вводе аргументов в окне функции кавычки ставить не нужно, т.к. они генерируются Мастером автоматически.

При вводе аргументов вручную непосредственно в строку формул – критерий следует заключать в кавычки.

10. Так как исходный диапазон совпадает с диапазоном суммирования, то 3-ий аргумент является необязательным – пропустите его. ОК.
11. Найдите в группе **Библиотека функций** категорию **Статистические**.
12. Для подсчета количества положительных чисел в диапазоне A2:J2 в ячейку M2 вставьте функцию **СЧЕТЕСЛИ()**.
13. Воспользовавшись справкой, примените функцию. Результат выполненного задания представлен на рис. 22.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1											Сумма	Ср. зн.	Сумм_если	Счит_если
2	-14	-11	-8	-5	-2	1	4	7	10	13	-5	-0,5	-40	5
3														

**Рис. 22. Результат выполнения задания 1**

14. Сохраните выполненное задание.

## Задание 2. Штраф за отходы

Рассчитать сумму штрафа для предприятия, которое каждый месяц сбрасывает в реку промышленные отходы.

Штраф взимается тогда, когда отходы превышают заданное предельное значение. При этом за каждый килограмм сверх нормы взимается фиксированная сумма.

Если предприятие в данный месяц сдает помещение в аренду, то штраф увеличивается в два раза, т.к. предприятие зарабатывает дополнительные средства, но не следит за своими арендаторами.

Требуется найти:

- в каких месяцах был начислен штраф;
- какой штраф был уплачен предприятием в каждом месяце;
- месяцы, когда был начислен максимальный штраф;
- сколько месяцев выплачивался штраф и суммарный штраф за год.

Исходными данными являются месяцы работы предприятия, количество отходов за каждый месяц, а также наличие арендованных помещений на предприятии.

Задание выполняется в файле *Заготовки Excel* на листе *Штраф за отходы*. Для контроля правильности выполнения задания на листе приведен образец расчета.

### Технология выполнения задания

1. Откройте файл *Заготовки Excel*, лист *Штраф за отходы*. Цветным фоном выделены ячейки, которые следует заполнить результатами расчетов.
2. В столбце D рассчитайте разность между отходами (столбец B) и предельным значением (ячейка C14).
3. В столбце E определите, начислен штраф в этом месяце или нет по следующему правилу: если превышение предельного значения больше 0, то штраф начислен (обозначим это 1), в противном случае – не начислен (обозначим это 0). Указанный результат можно получить с использованием логической функции ЕСЛИ.

Для ячейки E20 формула выглядит так:

=ЕСЛИ(D20>0;1;0)

4. В столбце F определите месяц, когда выплачивался штраф по следующему правилу: если в значение в ячейке столбца E равно 1 (т.е. штраф начислен), то дать ссылку на месяц (столбец A), в противном случае – ввести пустое значение (две кавычки) или пробел.

Для ячейки F20 формула выглядит так:

=ЕСЛИ(E20=1;A20;"")

5. В столбце G вычислите, какой был бы начислен штраф без учета аренды помещений по следующему правилу: если штраф начислен, то сумма штрафа равна произведению превышения отходов (столбец D) на штраф на 1 кг (ячейка C15).

Для ячейки G20 формула выглядит так:

=ЕСЛИ(E20=1;D20\*\$C\$15;0)

Значения столбца G являются вспомогательными для вычисления окончательной суммы штрафа с учетом аренды помещений.

6. В столбце H вычислите штраф с учетом аренды помещений по следующему правилу: если есть аренда (ячейка столбца C равна «да»), то сумма штрафа (столбец G) умножается на 2, в противном случае берется сумма штрафа из столбца G.

Для ячейки H20 формула выглядит так:

=ЕСЛИ(C20="да";G20\*2;G20).

Для вычисления окончательной суммы штрафа мы проверили два условия: наличие штрафа и наличие аренды. Для этого вычислили вспомогательный столбец G. Можно было бы получить итоговую сумму штрафа и без использования вспомогательного столбца. Тогда в одной формуле надо проверить сразу два условия. Это можно сделать при помощи двух вложенных функций ЕСЛИ().

Тогда для ячейки H20 формула выглядела бы так:

=ЕСЛИ(C20="нет";G20;ЕСЛИ(E20=1;D20\*\$C\$15\*2;0))

7. В ячейке H32 вычислите максимальный штраф с использованием встроенной функции МАКС():

=МАКС(H20:H31)

8. В столбце I вычислите месяц, в котором была уплачена максимальная сумма штрафа по следующему правилу: если значение в ячейке столбца H равно максимальному (H32), то надо указать месяц (из столбца A), в противном случае ввести пустое значение (две кавычки) или пробел.

Для ячейки I20 формула выглядит так:

=ЕСЛИ(H20=\$H\$32;A20;"")

9. В ячейке D35 вычислите, сколько месяцев выплачивался штраф (сумма по столбцу E).

10. В ячейке D36 вычислите общую сумму штрафа за год (сумма по столбцу H).

11. Сохраните выполненное задание.

### Задание 3. Фитнес-центр

Чтобы не потерять клиентов фитнес-центр предоставляет постоянным клиентам услуги со скидками. В рассматриваемой задаче фитнес-центр предлагает следующую систему скидок и правил их применения:

- скидки не суммируются, но начисляются по гибкой системе;
- клиенту может быть представлено до 3-х скидок, расположенных в таблице исходных данных в порядке приоритетности: 1-ая скидка действует на всю сумму абонеента, последующие - на сумму за вычетом уже предоставленных скидок;
- если фактическая сумма приобретенных услуг (за вычетом скидок) превышает 15000 руб. клиенту предоставляется дополнительный бесплатный пакет;
- если фактическая сумма приобретенных услуг (за вычетом скидок) превышает 25000 руб. покупателю предоставляется клубная карта.

Рассчитать стоимость услуг фитнес-центра с учетом предоставленных скидок, а также общий процент скидок. Выделить клиентов, имеющих право на

дополнительные услуги. Определить фамилии клиентов, которым предоставлена клубная карта.

Исходные данные представлены в виде списка клиентов (рис. 23) с указанием положенных им льгот («1» в соответствующей ячейке) и суммой приобретенных услуг без скидок.

ФИО	Начальное количество скидок	% скидок					
		Постоянный клиент	Утро, будние дни	День или фикс. время	Годовой абонемент	>=3 чел.	Льготная категория
Сергеева А.С.	0	1			1		
Яковлев И.А.	0		1		1		1
Дмитриева О.В.	0	1	1		1	1	
Николаева С.П.	0			1		1	1
Борисов А.А.	0	1		1	1		
Алексеев К.Э.	0		1				
Матвеева И.С.	0						1

Рис. 23. Исходные данные к заданию 4

Проценты скидок в таблице расположены в последовательных ячейках одной строки. Это даст возможность с помощью смешанных ссылок написать обобщенную формулу для одной скидки и распространить ее на столбцы с другими скидками.

Задание выполняется в файле *Заготовки Excel* на листе *Абонемент*. Результаты расчетов можно сверять с образцом, представленном на листе.

### Технология выполнения задания

1. Откройте файл *Заготовки Excel*, лист *Фитнес-центр*. Цветным фоном выделены ячейки, которые следует заполнить результатами расчетов.
2. Заполните столбец Сумма приобретенных услуг реальными значениями. Их можно найти в Интернете.
3. В столбце I определите возможное число скидок для каждого клиента. Для ячейки I16 формула выглядит так:  

$$=СУММ(C16:H16).$$
4. В столбце J рассчитайте количество скидок, которые могут быть предоставлены клиенту по следующему правилу: если количество скидок больше 3трех, то их количество ограничивается тремя, иначе остается без изменений. Для ячейки J16 формула выглядит так:  

$$=ЕСЛИ(I16>3;3;I16)$$
5. В ячейках диапазона L16:Q22 определите суммы услуг за вычетом предоставленных скидок по следующему правилу: если у клиента есть признак соответствующей скидки и количество предыдущих скидок <3, то предыдущая сумма приобретенных услуг уменьшается на процент скидки, иначе остается без изменения. Для ячейки L16 формула выглядит так:  

$$=ЕСЛИ(И(C16=1;СУММ($B16:B16)<3);K16*(1-C$15);K16).$$

В данной формуле используются смешанные ссылки, что дает возможность скопировать формулу без изменений во все ячейки диапазона L16:Q22.

6. В столбце R рассчитайте общий процент скидки как отношение сэкономленной суммы («общая сумма» – «окончательная сумма») к общей сумме услуг.
7. В столбце S отметьте признаком «1» клиентов, которым полагаются дополнительные услуги:  
Для ячейки S16 формула выглядит так:  
$$=ЕСЛИ(Q16>\$C\$11;1;"-")$$
8. В столбце T составьте список клиентов, которым следует оформить клубную карту. Для ячейки T16 формула выглядит так :  
$$=ЕСЛИ(Q16>\$C\$12;B16;"-").$$
9. Сохраните выполненное задание.

#### **Задание 4. Сравнение вкладов**

Произвести исследование доходности вкладов по шести банкам с учетом возможности пополнения.

Исходными данными являются годовой процент по вкладу, минимальная первоначальная сумма вклада, возможность пополнения (1 - «да» или 0 - «нет»), минимальная сумма пополнения по каждому банку. Вклад размещается на 2 года (24 месяца) с ежемесячной капитализацией<sup>1</sup>. Пополнение вклада возможно через 6 и 12 месяцев. Если сумма для пополнения не достаточна или пополнение банком не предусмотрено, сумма храниться в «дома» и прибавляется в конце рассматриваемого периода к сумме вклада.

Задание выполняется в файле *Заготовки Excel* на листе *Сравнение вкладов*.

#### **Технология выполнения задания**

1. Откройте файл *Заготовки Excel*, лист *Сравнение вкладов*. Цветом выделены ячейки, в которые следует ввести формулы для расчетов.
2. Для удобства дальнейший расчетов расположите суммы следующим образом:
  - в ячейке C19 имеющаяся сумма для вноса =K10;
  - в ячейке K19 сумма 1-го пополнения через 6 месяцев: =K11;
  - в ячейке T19 сумма 2-го пополнения через 12 месяцев: =K12.
3. В столбце C (Начальная сумма) определите, достаточна ли сумма в 3000 руб. (ячейка K10) для первоначального вклада в конкретный банк (C10:C15). Если сумма удовлетворяет условию – вклад оформляется (т.е. сумма вносится), если нет – в ячейку вводится 0.  
Для ячейки C21 формула выглядит так: =ЕСЛИ(\$C\$19>=C10;\$C\$19;0)
4. Для столбцов с D по I (месяцы хранения с 1-ого по 6-ой) определите значение суммы вклада после капитализации процентов, которая осуществляется в

---

<sup>1</sup> Капитализация – начисление процента не с первоначальной суммы вклада, а с вклада с начисленными процентами за предыдущий период.

конце предыдущего месяца. С учетом свойств смешанных ссылок введите обобщенную формулу расчета. Для ячейки D21 формула будет выглядеть так:  
 $=C21+C21*\$D10/12$ .

5. В столбце J определите возможность пополнения вклада по 1-ому сроку: если вклад пролежал меньше 24 месяцев, значение в ячейке равно 1, иначе – 0.

Напишите формулу так, чтобы ее можно было скопировать для определения возможности пополнения вклада и по 2-му сроку (столбец S).

6. В столбце K определите сумму 1-ого пополнения вклада: если возможность пополнения есть (J21=1) И предполагаемая сумма пополнения (K19)>= минимальной сумме пополнений, установленной данным банком, значение в ячейке равно K19, иначе – 0.

Для ячейки K21 формула выглядит так:

$$=ЕСЛИ(И(J21=1;K\$19>=\$E10);K\$19;0)$$

Формула составлена так, что ее можно с помощью копирования использовать и для определения суммы 2-ого пополнения вклада (столбец T). Этому также способствует и соответствующее расположение исходных данных.

7. В столбце L подсчитайте сумму с учетом пополнения (сумма значений в столбцах I и K).

8. Скопируйте формулы диапазона D21: L26 в диапазон M21:U26. Таким образом, вы получите данные о росте вклада с 7-ого по 12-ый месяцы хранения, возможность пополнения вклада по 2-му сроку, сумму 2-ого пополнения вклада и скорректированную сумму 12.

9. Для столбцов с V по AG (Месяцы хранения с 13-ого по 24-ый) определите сумму ежемесячного прироста вклада. Для этого скопируйте формулу из ячейки M21 в ячейку V21 и распространите ее на весь диапазон.

10. В столбце AH определите сумму, накопленную вкладчиком с учетом тех сумм, которые хранились дома, т.к. некоторые банки не предоставили услугу пополнения. Для этого к сумме, находящейся в банке по окончании срока хранения (ячейка AG21) следует добавить разницу между суммой, которую вкладчик намеревался сохранить и суммой, которую принял банк с учетом ограничений.

Для ячейки AH21 формула выглядит так:

$$=AG21+\$C\$19+\$K\$19+\$T\$19-C21-K21-T21.$$

11. В столбце AI определите процент реального прироста денежных средств вкладчика следующим образом: из окончательной суммы вычитаем первоначальный взнос и суммы пополнений, и полученный прирост делим на окончательную сумму. Для ячейки AI21 формула выглядит так:

$$=(AH21-(\$C\$19+\$K\$19+\$T\$19))/AH21$$

Формат ячеек – процентный.

12. В столбец AJ для наглядности сравнения обещанного и полученного прироста перенесите данные из D10:D15.

13. В столбце АК найдите разницу между реальным приростом денежной массы в кошельке вкладчика и приростом, обещанным банком.

14. Сохраните выполненное задание.